
Задание:

Необходимо написать программу, реализующую параллельную работу нескольких процессов. Каждый процесс может состоять из одного или нескольких потоков. Любой из потоков, работающих в составе этих процессов, может быть приостановлен и вновь запущен некоторой определенной клавишей (можно использовать буквенные или цифровые клавиши). Нажатия клавиш обрабатывать с помощью прерывания от клавиатуры (по материалам лаб. работы №1).

Окончание работы программы должно происходить при приостановке всех потоков их ключевыми клавишами либо при нажатии клавиши ESC. При окончании работы необходимо выполнить корректное завершение, т.е. «дочитать» всю информацию из буфера каждого процесса (при его наличии), закрыть все открытые файлы и т.п. – по материалам лаб. работы №3.

Те потоки, которые выводят информацию на экран, должны использовать для этого каждый свое отдельное окно, обрамленное рамкой – экран визуально делится на несколько окон, в каждом из которых отображается один процесс.

Дополнительное информационное окно должно содержать описание управляющих клавиш и краткое сообщение о тех действиях, которые выполняются в программе: номер потока, его статус в текущий момент времени (активен, приостановлен, ожидает), другую информацию (указание величины выделенного потоку кванта времени, процент заполненности буфера и т.п.) в зависимости от конкретного потока.

Программа должна работать устойчиво, без «зависаний» и непредвиденных зрительных эффектов. В случае необходимости указания параметров при запуске программы необходимо предусмотреть и обработать все возможные ошибки (с выводом диагностики на экран).

Задачи 1, 3, 6 предполагают наличие управляющего блока, который, используя прерывания таймера, случайным образом определяет очередной активный поток и выделяет ему кванты времени (возможно, в рамках большего кванта, выделенного всему процессу). Кванты времени – как для процесса в целом, так и для его потоков – изначально задавать некоторой фиксированной величины, но предусмотреть возможность её изменения во время работы путем нажатия некоторых ключевых клавиш (для каждого из потоков предусмотреть свою клавишу) – т.е. можно, например, увеличить или уменьшить квант только потока-производителя. При этом допустимы разные варианты реализации – «общий» квант может либо меняться, либо оставаться постоянным (тогда при ускорении одного потока другой автоматически замедлится, т.к. величина его кванта уменьшится).

Потоки этого класса задач могут иметь три статуса: «активен», «ожидает» или «приостановлен». В процессе работы может возникнуть, например, следующая ситуация. Поток-потребитель приостановлен своей ключевой клавишей, следовательно, буфер не освобождается. Поток-производитель активен, он заполнит буфер информацией и перейдёт в состояние ожидания. Из этого состояния он сможет выйти только после того, как будет возобновлена (нажатием клавиши) работа потока-потребителя, который освободит место в буфере для помещения новой информации. В случае приостановки производителя возникнет аналогичная ситуация, только с пустым буфером. Приостановка/возобновление потока возможны в любом его состоянии – как в активном, так и в состоянии ожидания.

Задачи 2, 4, 5, 7 должны выполняться равномерно, независимо от степени загрузки системы. Для этого каждой из них необходимо получать управление через фиксированное количество «тиков» системного таймера, во время которого они выполнят какое-то свое элементарное действие («бегущая строка» или «летающий объект» сместится на одну позицию, сменится нота в музыке…). При такой реализации скорость каждого потока будет определяться количеством «тиков» таймера между его запусками. Для уменьшения скорости такого потока достаточно после нажатия ключевой клавиши предоставлять ему управление реже, через большее число «тиков», соответственно для ускорения такого потока – опять же после нажатия ключевой клавиши – ему предоставляется управление чаще, в пределе – на каждом «тике». Так, «бегущая строка» должна двигаться равномерно с постоянной скоростью (если она не приостановлена ключевой клавишей) независимо от количества активных процессов в системе, музыка – тоже играть равномерно… Потоки этого класса задач могут иметь два статуса: «активен» и «приостановлен».

Задачи:

1. Два потока: первый читает информацию из файла (например, стихи или текст программы) в буфер, второй эту информацию из буфера выдаёт на экран. При заполнении окна вывода до конца его содержимое не должно обновляться полностью – вывод новой информации должен осуществляться в последнюю строку, а все остальные строки смещаться вверх (по материалам лаб. работы №3). Имя читаемого файла задавать как параметр командной строки. После окончания файла он начинает считываться заново.

2. Музыкальное сопровождение – как минимум несколько нот различной длительности, образующие мелодию. Во включённом состоянии мелодия должна играть в фоновом режиме (по материалам лаб. работы №2).

3. Два потока: один выполняет поиск всех последовательных простых чисел и заносит их в буфер, второй осуществляет их вывод из буфера на экран (по материалам лаб. работы №3).

4. Вывод заголовка работы (фамилия и имя автора и название работы) постоянно меняющегося цвета. Место вывода (верхняя или нижняя строка экрана) задавать в качестве параметра командной строки.

5. Вывод заголовка работы (фамилия и имя автора и название работы) в виде «бегущей строки». Место вывода (верхняя или нижняя строка экрана) задавать параметром командной строки.

6. Два потока: один генерирует случайные числа и выбирает в буфер те, которые являются полным квадратом, второй выводит их из буфера на экран (по материалам лаб. раб. №3).

7. Движущийся объект (в простейшем случае, например, летающий и отражающийся от границ окна шарик). Движение может осуществляться свободно или при управлении с клавиатуры.

8. В углу экрана с заданными в качестве параметров координатами поместить часы, показания которых должны соответствовать системному времени. Обновлять показания часов каждую секунду (по материалам лаб. работы №2).

Задания:

3. Два потока: один выполняет поиск всех последовательных простых чисел и заносит их в буфер, второй осуществляет их вывод из буфера на экран (по материалам лаб. работы №4).

5. Вывод заголовка работы (фамилия и имя автора и название работы) в виде “бегущей строки” (по материалам лаб. работы №5). Место вывода (верхняя или нижняя строка экрана) задавать параметром командной строки.

7. Движущийся объект (в простейшем случае, например, летающий и отражающийся от границ окна шарик). Движение может осуществляться свободно или при управлении с клавиатуры (по материалам лаб. работы №5).

8. В углу экрана с заданными в качестве параметров координатами поместить часы, показания которых должны соответствовать системному времени. Обновлять показания часов каждую секунду (по материалам лаб. работы №3).

Сведения об интерфейсе программы и детали его начальной реализации.

Реализация окон для процессов.
При работе программы основное место на экране выделено под информации о потоках, которые ищут простые числа и выводят (задание 3). 

Процессу, реализующему задание 5, отведены верхняя или нижняя строки экрана. 

Процесс, реализующий задание 7, не нуждается в специальной области, т.к. шарик (изображается розовым символом “O”) может двигаться по всему экрану. 

Процессу, реализующему задание 8, отведены четыре угла экрана (верхний левый, верхний правый, нижний левый или нижний правый). 

Подсказки, описывающие управляющие клавиши программы, находятся в нижней части экрана под основным окном процесса задания 3.

Общий вид экрана имеет следующий вид:

Потоки программы

Процесс движения бегущей строки может быть остановлен/запущен нажатием клавиши F3. Скорость движения бегущей строки может увеличиваться клавишей 4 и уменьшаться клавишей 5. Большее значение скорости задает более быстрое движение бегущей строки.

Процесс обновления показания часов каждую секунду остановлен/запущен нажатием клавиши F4.

Остановка/запуск движения шарика выполняется нажатием клавиши F5. Задержка между сменами положения двигающегося символа может увеличиваться клавишей 6 и уменьшаться клавишей 7. (Скорость движения символа добавляется клавишей 7 и уменьшается клавишей 6). Данная задержка задается в единицах системного таймера, равных 55 мс.

Состояние потоков задания 3 могут принимать одно из трех значений («Активен»/«Ожидает»/«Приостановлен»). Текущее состояние каждого потока отображаются на экране. Для управления состояниями потоков «Активен»/«Приостановлен» используются клавиши F1 и F2, в состояние «Ожидает» поток переходит автоматически при возникновении определенных условий.

Для изменения квантов времени, выделяемым потокам, используются клавиши 0,1,2,3. 
Параметры командной строки.

Согласно заданию курсовой работы, в программу передаются два параметра:

-положение часов;

-положение бегущей строки.

Положение часов задается символами 1 (верхний левый угол экрана), 2 (верхний правый угол экрана), 3 (нижний левый угол экрана), 4 (нижний правый угол экрана).

Положение бегущей строки задается символом 0 (верхняя строка) или 1 (нижняя строка).

Пример:

kurs_20.exe 1 1

запускает программу с бегущей строкой в нижней строке, часы отображаются в верхнем левом углу.

Если параметры не будут указан или указаны не полностью, то программа закончит работу, выдав следующее сообщение:

Описание основных алгоритмов, используемых в программе.

Для поиска простых чисел используется последовательный перебор всех целых чисел с проверкой каждого числа на простое число. Описание алгоритма: Мы будем проверять каждое число на деление его без остатка на числа из последовательности от 2 до целой части от квадратного корня исходного числа. Например, для числа 17 это будут числа 2,3,4. Понятно, что после неудачной проверки деления на 2 проверять деление на остальные четные числа смысла не имеет, но про эту оптимизацию чуть позже. Если остаток от деления будет равен нулю, то это означает, что число не простое, и мы завершаем проверку.  При неудачной проверке берется число большее текущего на единицу. Найденное простое число заносится в буфер. Если в буфере нет свободного места, то поток переходит в состояние ожидания и новых простых чисел не ищет. Поиск простых чисел продолжается до тех пор, пока не истечет время, отведенное на работу потока поиска. Поток печати простых чисел проверяет буфер на пустоту и, если буфер не пуст, то извлекает первое число из буфера и выводит его на экран. Если буфер пуст, то поток переходит в режим ожидания. Печать простых чисел происходит до истечения времени, отведенного потоку. При каждом изменении элементов в буфере происходит вывод на экран процента заполнения буфера.
Бегущая строка движется справа налево. При движении анализируется положение часов на экране, чтобы не затереть их показания. 

Шарик движется по всему экрану, отражаясь от его границ. При каждой смене положения, выполняется сохранение и восстановление содержимого экрана под шариком. Задержка между сменами положения шарика задается в единицах системного таймера, равных 55 мс.
Описание основных переменных, констант и типов.

В программе используются константы:

RunningLine – текст бегущей строки;

TDELAY – задержка после каждого вызова процедур потоков (в мс).

Глобальные переменные.

IBuffer  – буфер простых чисел;

ITop – индекс последнего элемента в IBuffer;

StPNP, StPNS – состояния потоков;

SaveInt8, SaveInt9 – процедуры (имеют тип Procedure) используемые при хранения обработчиков прерываний таймера и клавиатуры;

FC – состояние процесса смены часов;

FB – состояние процесса движения шарика по экрану;

FR – состояние процесса движения бегущей строки;

TBall – задержка между движениями шарика;

XBall, YBall – координаты двигающегося шарика;

Prime – текущее простое число;

RunningSpeed – скорость движения бегущей строки;

XRunning, YRunning – координаты бегущей строки;

Screen – массив для прямого доступа к видеопамяти;

Программная реализация.

Программа написана на Borland Pascal 7.

В тексте программы в разделе описания переменных приведены комментарии для основных переменных и процедур.

Для обработки прерываний клавиатуры используется процедура IntProcKeyboard, для обработки прерываний таймера – IntProc.

OutBall – процедура вывода на экран двигающегося шарика.

OutRunningLine – процедура сдвига и выводы на экран бегущей строки.

OutLine – процедура вывода символьной строки в видеопамять.

OutProgress – процедура вывода процента заполнения буфера простых чисел.

Результаты работы программы

Приведен вид экрана в момент работы программы:

Исходный текст программы.

program kurs_20;

{$M $1000,0,0}

uses CRT, DOS;

type
 Status=(Assigned,Wait,Suspended); {состояния потоков}

const

 RunningLine: String='Миронов В.А.';

 TDELAY=15;

var

 Screen: Array[1..25,1..80,1..2] of Byte Absolute $B800:$0000; {видеопамять экрана}

 IBuffer: Array[1..15] of Longint; {буфер простых чисел}

 S: String;

 SaveInt8, SaveInt9: Procedure; {обработчики прерываний}

 Prime: Longint;

 Hour, Minute, Second: Word; {текущее время}

 T1000: Word; {время в мс, прошедшее со времени изменения значения секунд}

 XClock, YClock: Integer;

 FC: Boolean;

 ABall, CBall: Byte; {символ и атрибут экрана под двигающимся символом}

 DTBall: Integer; {число тиков таймера от последнего движения}

 TBall: Integer; {задержка между перемещения символа [в тиках таймера]}

 DXBall, DYBall, XBall, YBall: Integer;

 FB: Boolean;

 CX, CY, ITop: Integer;

 RunningSpeed, TRunning: Word;

 XRunning, YRunning: Integer;

 FR: Boolean;

 FClose: Boolean;

 StPNP, StPNS: Status;

 RW_CTime, PN_CTime, time: Word;

 PWET, PRET: Integer;

function LeadingZero(V: Word): String; {преобразование числа в двухсимвольную строку}

var S: String;

begin

 Str(V:0,S);

 if (V<10) then S:='0'+S;

 LeadingZero:=S;

end;

procedure OuTBall; {вывод двигающегося символа}

var I, L, X, X1, X2: Integer;

begin
 Screen[YBall,XBall,2]:=ABall; {восстановление прежнего содержимого экрана}

 screen[YBall,XBall,1]:=CBall;

 Inc(XBall,DXBall);

 Inc(YBall,DYBall);

 if XBall<=1 then begin

  DXBall:=-DXBall;

  XBall:=2-XBall;

 end

 else

  if XBall>=80 then begin

   DXBall:=-DXBall;

   XBall:=160-XBall;

  end;

 if YBall<=1 then begin

  DYBall:=-DYBall;

  YBall:=2-YBall;

 end

 else

  if YBall>=25 then begin

   DYBall:=-DYBall;

   YBall:=50-YBall;

  end;

 ABall:=Screen[YBall,XBall,2]; {cохранение прежнего содержимого экрана}

 CBall:=Screen[YBall,XBall,1];

 Screen[YBall,XBall,2]:=MAGENTA;

 Screen[YBall,XBall,1]:=Byte('O');

end;

procedure OutLine(S: String; A: Byte; X, Y: Integer); {вывод строки в видеопамять}

var I: Integer;

begin

 I:=1;

 while ((X<=80) And (I<=Length(S))) do begin

  if (X=XBall) And (Y=YBall) then begin

   ABall:=A;

   CBall:=Byte(S[i]);

  end

  else begin

   Screen[Y,X,2]:=A;

   Screen[Y,X,1]:=Byte(S[i]);

  end;

  Screen[Y,X,2]:=A;

  Screen[Y,X,1]:=Byte(S[I]);

  Inc(I);

  Inc(X);

 end;

end;

procedure SetCursor(Cursor: Word); {установка типа курсора}

var  Regs: Registers;

Begin

 with Regs do begin

  AH:=1;

  BH:=0;

  CH:=Hi(Cursor);

  CL:=Lo(Cursor);

 end;

 Intr($10,Regs);

end;

procedure OutProgress(MaxSize, Size, X, Y: Integer); {вывод % заполнения буфера}

var S: String;  V: Real;

begin

 V:=Size/MaxSize*100;

 Str(V:2:2,S);

 OutLine('Заполняемость буфера:',WHITE,X,Y);

 OutLine(S+' % ',WHITE,X+22,Y);

end;

function PrimeNumber(V: Longint): Boolean; {проверка является число простым}

var I: Integer;

begin

 PrimeNumber:=FALSE;

 if (V<=1) then exit;

 for I:=2 to Trunc(Sqrt(V)) do

  if ((V Mod I)=0) then exit;

 PrimeNumber:=TRUE;

end;

procedure PrimeNumberSearch; {поиск простых чисел и занесение их в буфер}

var S, ThreadTime: String;

begin

 case StPNS of {вывод состояния потока}

Assigned:  OutLine('Активен    ',WHITE,40,3);

Wait:      OutLine('Ожидание   ',WHITE,40,3);

Suspended: begin OutLine('Остановлен ',WHITE,40,3); exit; end;

 end;

 if ITop>=15 then begin {буфер полон}

  PN_CTime:=65535;

  StPNS := Wait;       {переход в режим ожидания}

  OutLine('Ожидание   ',WHITE,40,3); {вывод нового состояния}

  exit;

 end;

 if StPNS = Wait then begin {переход в активный режим, если ждали}

  StPNS := Assigned;

  OutLine('Активен    ',WHITE,40,3); {вывод нового состояния}

 end;

 repeat  {поиск следующего простого числа}

  Inc(Prime);

 until PrimeNumber(Prime) Or (Prime>=MAXLONGINT);

 if Prime>=MAXLONGINT then exit; {достигнут предел чисел типа Longint}

 Inc(ITop); {добавляем число в буфер}

 IBuffer[ITop]:=Prime;

 OutProgress(15,ITop,25,5);

 Delay(TDELAY);

end;

procedure PrimeNumberPrint; {вывод простых чисел из буфера}

var

  i: Integer;

  X, Y: Integer;

  tmp, S: String;

begin

 case StPNP of {вывод состояния потока}

Assigned:  OutLine('Активен    ',WHITE,40,4);

Wait:      OutLine('Ожидание   ',WHITE,40,4);

Suspended: begin OutLine('Остановлен ',WHITE,40,4); exit; end;

 end;

 if ITop<=0 then begin

  PN_CTime:=65535;

  StPNP:=Wait;

  OutLine('Ожидание   ',WHITE, 40, 4);

  exit;

 end;

 if StPNP = Wait then begin {переход в активный режим, если ждали}

  OutLine('Активен    ',WHITE, 40, 4);

  StPNP:=Assigned;

 end;

 Str(IBuffer[1], tmp);

 if (CX+Length(tmp)+1>75) then begin  {число не помещается в текущей строке}

  CX:=2; {переход на следующую}

  Inc(CY);

 end;

 if CY > 13 then begin {следующей строки нет - очищаем область окна и пишем по новому}

  CY:=7;

  TextBackGround(BLACK);

  Window(5,7,78,16);

  ClrScr;

 end;

 OutLine(tmp+' ',WHITE,5+CX,CY);

 CX := CX + Length(tmp) + 1;

 Dec(ITop);

 for I:=1 to ITop do IBuffer[I]:=IBuffer[I+1]; {удаление первого простого числа из буфера}

 OutProgress(15,ITop,25,5);

 Delay(TDELAY);

end;

procedure OutRunningLine; {вывод бегущей строки в видеопамять}

var I, L, X, X1, X2: Integer;

begin

 if (XRunning>1) then Dec(XRunning)

                 else XRunning:=80;

 if (YRunning=YClock) then

  if (XClock<40) then begin

   X1:=XClock+8; X2:=80;

  end

  else begin

   X1:=1; X2:=XClock-1;

  end

 else begin

  X1:=1; X2:=80;

 end;

 for X:=X1 to X2 do begin {стирание строки пробелами}

  Screen[YRunning,X,1]:=32;

  Screen[YRunning,X,2]:=BLACK;

 end;

 L:=Length(RunningLine);

 if (L>80) then L:=80;

 X:=XRunning;

 for I:=1 to L do begin

  if ((X>=X1) And (X<=X2)) then begin

   Screen[YRunning,X,2]:=LIGHTBLUE;

   Screen[YRunning,X,1]:=Byte(RunningLine[I]);

  end;

  Inc(X);

  if (X>=80) then X:=1;

 end;

end;

{$F+}

procedure IntProc; {обработка прерываний системного таймера }

Interrupt;

begin

 Inc(PN_CTime);

 Inc(RW_CTime);

 Inc(T1000,55);

 if (T1000>=1000) then begin

  Dec(T1000,1000);

  Inc(Second);

  if (Second>=60) then begin

   Second:=0;

   Inc(Minute);

   if (Minute>=60) then begin

    Minute:=0;

    Inc(Hour);

    if (Hour>=24) then Hour:=0;

   end;

  end;

  if FC then {процесс "часы" активен}

   OutLine(LeadingZero(Hour)+':'+LeadingZero(Minute)+':'+LeadingZero(Second),WHITE,XClock,YClock);

 end;

 if FB then begin {процесс "двигающийся символ" активен}

  inc(DTBall);

  if DTBall>=TBall then begin

   DTBall:=0;

   OuTBall;

  end;

 end;

 if FR then begin {процесс "бегущая строка" не остановлен}

  Inc(TRunning);

  if (TRunning>Round(18.2/RunningSpeed)) then begin

   TRunning:=0;

   OutRunningLine;

  end;

 end;

 Inline($9C);

 SaveInt8;

end;

{$F-}

procedure Quit; {завершение работы программы}

begin

 SetIntVec($8,Addr(SaveInt8));

 SetIntVec($9,Addr(SaveInt9));

 StPNP:=Assigned;

 time:=0;

 PWET:=0;

 while (ITop>0) do PrimeNumberPrint;

 Window(1,1,80,25);

 TextBackGround(BLACK);

 ClrScr;

 SetCursor($0607); {установка обычного курсора}

 Halt(0);

end;

{$F+}

procedure IntProcKeyboard; {обработка прерываний клавиатуры}

Interrupt;

begin

 case Port[$60] of

1:  FClose:=TRUE;




{нажат Esc}

11: if PRET<10 then Inc(PRET);



{нажат 0}

2:  if PRET>0 then Dec(PRET);



{нажат 1}

3:  if PWET<10 then Inc(PWET);



{нажат 2}

4:  if PWET>0 then Dec(PWET);



{нажат 3}

5:  if RunningSpeed<16 then begin


{нажат 4}

     Inc(RunningSpeed);

     Str(RunningSpeed:3,S);

     OutLine('Скорость бегущей строки   '+S+' [1-16]',YELLOW,25,24);

    end;

6: if RunningSpeed>1 then begin



{нажат 5}

     Dec(RunningSpeed);

     Str(RunningSpeed:3,S);

     OutLine('Скорость бегущей строки   '+S+' [1-16]',YELLOW,25,24);

    end;

7:  if TBall<50 then begin



{нажат 6}

     Inc(TBall);

     Str(TBall:3,s);

     OutLine('Задержка движения символа '+S+' [1-50]',YELLOW,25,23);

    end;

8:  if TBall>1 then begin



{нажат 7}

     Dec(TBall);

     Str(TBall:3,S);

     OutLine('Задержка движения символа '+S+' [1-50]',YELLOW,25,23);

    end;

59: if StPNS<>Suspended then StPNS:=Suspended else StPNS:=Assigned;
{нажат F1}

60: if StPNP<>Suspended then StPNP:=Suspended else StPNP:=Assigned;
{нажат F2}

61: FR:=Not FR;







{нажат F3}

62: FC:=Not FC;







{нажат F4}

63: FB:=Not FB;







{нажат F4}

 end;

 Inline($9C);

 SaveInt9;

end;

{$F-}

begin

 if (ParamCount<>2) then begin

  Writeln; Writeln('Ошибка в параметрах');

  Writeln; Writeln('Параметры программы:');

  Writeln; Writeln('kurs_20.exe c s');

  Writeln('c - положение часов: 1 - верхний левый угол, 2 - верхний правый угол');

  Writeln('                     2 - нижний левый угол, 4 - нижний правый угол');

  Writeln('s - положение бегущей строки: 0 - верхняя строка экрана, 1 - нижняя');

  Halt;

 end;

 S:=ParamStr(1);

 if (S<>'1') And (S<>'2')  And (S<>'3') And (S<>'4') then begin

  Writeln; Writeln('Ошибка в параметрах');

  Writeln('Положение часов должно быть 1, 2, 3 или 4');

  Halt;

 end;

 if S='1' then begin  XClock:=1; YClock:=1; end

 else

  if S='2' then begin XClock:=71; YClock:=1; end

  else

   if S='3' then begin XClock:=1; YClock:=25; end


    else begin XClock:=71; YClock:=25; end;

 S:=ParamStr(2);

 if (S<>'0') And (S<>'1') then begin

  Writeln; Writeln('Ошибка в параметрах');

  Writeln('Положение бегущей строки должно быть 0 или 1');

  Halt;

 end;

 if S='0' then YRunning:=1

          else YRunning:=25;

 Randomize;

 TextBackGround(BLACK); ClrScr;

 Window(1,2,80,17); TextBackGround(BLUE); ClrScr;

 Window(2,3,79,16); TextBackGround(BLACK); ClrScr;

 TextColor(WHITE);

 Window(1,1,80,25); 

 GotoXY(25,2);

 Write('Процесс работы с простыми числами');

 OutLine('Поток 1:',WHITE,30,3);

 OutLine('Поток 2:',WHITE,30,4);

 Window(2,18,79,24);

 Writeln('F1 - Старт/Стоп потока поиска      F2 - Старт/Стоп потока вывода');

 Writeln('F3 - Старт/Стоп движения строки    F4 - Старт/Стоп часов');

 Writeln('F5 - Старт/Стоп движения символа');

 Writeln('0/1 - +/- время потока поиска      2/3 - +/- время потока вывода');

 Writeln('4/5 - +/- скорости бегущей строки, 7/6 - +/- скорости движения символа');

 Writeln('Esc - выход');

 FClose:=FALSE;

 GetTime(Hour,Minute,Second,T1000);

 T1000:=T1000*10;

 TBall:=12;

 DTBall:=0;

 if Random(2)=1 then DXBall:=1 else DXBall:=-1;

 if Random(2)=1 then DYBall:=1 else DYBall:=-1;

 XBall:=40+Random(20);

 YBall:=12+Random(10);

 FB:=TRUE;

 FC:=TRUE;

 FR:=TRUE;

 ITop:=0;

 Prime:=1;

 RunningSpeed:=12;

 TRunning:=0;

 XRunning:=81;

 StPNS:=Assigned;

 StPNP:=Wait;

 CX:=2; CY:=7;

 PWET:=0; PRET:=0;

 ABall:=Screen[YBall,XBall,2];

 CBall:=Screen[YBall,XBall,1];

 OuTBall;

 GetIntVec($8,@SaveInt8);

 SetIntVec($8,Addr(IntProc));

 GetIntVec($9,@SaveInt9);

 SetIntVec($9,Addr(IntProcKeyboard));

 Str(TBall:3,S);

 OutLine('Задержка движения символа '+S+' [1-50]',YELLOW,25,23);

 Str(RunningSpeed:3,S);

 OutLine('Скорость бегущей строки   '+S+' [1-16]',YELLOW,25,24);

 SetCursor($2000); {отключение курсора}

 while True do begin

  if FClose then Quit;

  time:=Random(3)+1;

  PN_CTime:=0;

  RW_CTime:=0;

  if Random(2)=0 then begin

   Str(time+PWET:3,S);

   OutLine('T '+S,Yellow,50,3);

   while (PN_CTime<time+PWET) do PrimeNumberSearch;

   if StPNS=Assigned then OutLine('Время вышло   ',WHITE,40,3);

  end

  else begin

   Str(time+PRET:3,S);

   OutLine('T '+S,Yellow,50,4);

   while (PN_CTime<time+PRET) do PrimeNumberPrint;

   if StPNP=Assigned then OutLine('Время вышло   ',WHITE,40,4);

  end;

 end;

end.

